
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 11.1.3.4 which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_11_1_3_4
Test Group:
/11/UTRAN34_SM_NIMTC.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D15wk25 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel 7260
Verification Status:
PASS
4 Corrections required for test case 11.1.3.4
4.1 Introduction

This section describes the changes required to make test case 11.1.3.4 run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D15wk25 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 Module UTRAN34_SM_NIMTC
	Record Name
	UTRAN34_SM_NIMTC

	Reason for change
	In main testcase body when the RB setup is done, it is done on basis of the accessStratumReleaseIndicator received in rrcConnectionREQ (at the start of testcase). This needs to be used to set the testcase release. For this the function fl_SetTestCaseRelease is used which is defined in module UTRAN34_UE_CapabilityInfo. This module needs to be imported in module UTRAN34_SM_NIMTC, which is not done in current TTCN implementation..

	Summary of change
	Imported module UTRAN34_UE_CapabilityInfo in UTRAN34_SM_NIMTC

	Source of change
	UTRAN34_SM_NIMTC.ttcn

	
	


Before:

	module UTRAN34_SM_NIMTC
{
  import from CommonDefs all;
  import from UTRAN_CommonDefs all;
  import from UTRAN_Component all;
  import from UTRAN_TypeDefs all;
  import from UTRAN_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from UTRAN_ASP_Definitions language "ASN.1:2002" all;
  import from NAS_24008TypeDefs all;
  import from NAS_CommonTypeDefs all;
  import from NAS_CommonTemplates all;
  import from Parameters all;
  import from UTRAN_TTCN_ASP_Templates all;
  import from UTRAN_RAB_Functions all;
  import from UTRAN34_ConfigurationSteps all;
  import from UTRAN34_CellInfo all;
  import from UTRAN34_CommonFunctions all;
  import from UTRAN34_NAS_PDU_Templates all;
  import from UTRAN34_RRC_CommonFunctions all;
  import from UpperTesterFunctions all;
  import from UTRAN_RRC_Templates all;
  import from UTRAN_ASN1_ASP_Templates all;
  import from UTRAN_CommonFunctions all;
  import from UTRAN34_PDCP_Functions all;
  


After:

	module UTRAN34_SM_NIMTC
{
  import from CommonDefs all;
  import from UTRAN_CommonDefs all;
  import from UTRAN_Component all;
  import from UTRAN_TypeDefs all;
  import from UTRAN_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from UTRAN_ASP_Definitions language "ASN.1:2002" all;
  import from NAS_24008TypeDefs all;
  import from NAS_CommonTypeDefs all;
  import from NAS_CommonTemplates all;
  import from Parameters all;
  import from UTRAN_TTCN_ASP_Templates all;
  import from UTRAN_RAB_Functions all;
  import from UTRAN34_ConfigurationSteps all;
  import from UTRAN34_CellInfo all;
  import from UTRAN34_CommonFunctions all;
  import from UTRAN34_NAS_PDU_Templates all;
  import from UTRAN34_RRC_CommonFunctions all;
  import from UpperTesterFunctions all;
  import from UTRAN_RRC_Templates all;
  import from UTRAN_ASN1_ASP_Templates all;
  import from UTRAN_CommonFunctions all;
  import from UTRAN34_PDCP_Functions all;
  import from UTRAN34_UE_CapabilityInfo all; //WA#11_1_3_4


4.3 f_TC_11_1_3_4_UTRAN()
	Record Name
	f_TC_11_1_3_4_UTRAN()

	Reason for change
	In the curent TTCN implementation, the function f_UTRAN34_Preamble is used to achieve the initial condition. This function expects the establishment cause set to registration for the rrc connection request. 

This is incorrect.

As per 24.008 Annex L, table L.1.2, the RRC Establishment cause for Any PS NAS procedure where the initiating uplink signalling message has the Device properties IE with low priority indicator set to "MS is configured for NAS signalling low priority" will be set as delayTolerantAccess
This needs to be changed.

	Summary of change
	Since the function f_UTRAN34_Preamble is used in many other places (for testcase other than NIMTC), created a new function f_UTRAN34_Preamble_NIMTC and called it instead.

	Source of change
	UTRAN34_SM_NIMTC.ttcn

	
	


Before:
	
function f_TC_11_1_3_4_UTRAN() runs on UTRAN_PTC
  {
    //Initialise UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    f_UT_USIM_Insert(UT, "34.123-1 / 11.1.3.4");
    //Idle update to bring UE to Cell_DCH state to register and to release connection again
    f_UTRAN34_Preamble(utran34_CellA);

    f_UTRAN_TestBody_Set(true);
    fl_TC_11_1_3_4_Body();
    f_UTRAN_TestBody_Set(false);
    //Release RRC Connection
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //Release UTRAN Cell
    f_UTRAN34_Postamble(utran34_CellA, U1_IDLE);
  }


After:

	function f_TC_11_1_3_4_UTRAN() runs on UTRAN_PTC
  {
    //Initialise UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    f_UT_USIM_Insert(UT, "34.123-1 / 11.1.3.4");
    //Idle update to bring UE to Cell_DCH state to register and to release connection again
    //WA#11_1_3_4 f_UTRAN34_Preamble(utran34_CellA);
    f_UTRAN34_Preamble_NIMTC (utran34_CellA);//WA#11_1_3_4
    f_UTRAN_TestBody_Set(true);
    fl_TC_11_1_3_4_Body();
    f_UTRAN_TestBody_Set(false);
    //Release RRC Connection
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //Release UTRAN Cell
    f_UTRAN34_Postamble(utran34_CellA, U1_IDLE);
  }


4.4 F_UTRAN34_Preamble_NIMTC
	Record Name
	f_UTRAN34_Preamble_NIMTC

	Reason for change
	Please see change 4.3 above

	Summary of change
	New function

	Source of change
	UTRAN34_CommonFunctions.ttcn

	
	


Before:
	function f_UTRAN34_Preamble_NIMTC(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  { /* @sic R5s130347: local GuardTimer sic@ */
    timer t_GuardTimer := tsc_GuardTimePreamble;
    t_GuardTimer.start;
    //  Switch cell power on
    f_UTRAN_SetCellPower(p_CellId, tsc_AttenuationServingCell);
    //  Switch on UE
    f_UT_SwitchOnUE(UT, false);
    //  Establish RRC Connection, perform Authentication, Security Mode procedure,
    //  Attach procedure and release RRC Connection
    f_UTRAN34_IdleUpdated(p_CellId, omit, delayTolerantAccess);
    t_GuardTimer.stop;
  }


4.5 fl_TC_11_1_3_4_Body()
	Record Name
	fl_TC_11_1_3_4_Body()

	Reason for change
	In main testcase body when the RB setup is done, it is done on basis of the accessStratumReleaseIndicator received in rrcConnectionREQ (at the start of testcase). This needs to be used to set the testcase release.
In current TTCN implementation, this is not being done.

Later in the testcase, the implementation of steps 4,5 and 6 is not correct. The current TTCN implementation starts an upperbound and a lowerbound timer, upperbound for 10s plus 10% tolerance, and lowerbound for 10s minus 10% tolerance. As soon as the lowerbound times out, the testcase expects the UE to send a rrc connection request for the service request procedure which does the PDP context activation(Steps 5 and 6). 

This is also incorrect. The UE will not trigger a rrcConnectionREq by itself. It will need an AT command to trigger this which is not implemented in current TTCN.

This needs to be implemented.

	Summary of change
	Called the function fl_SetTestCaseRelease to set the release for the testcase.

Disabled the implementation of Steps 4,5 and 6 (in blue), and implemented the AT command to trigger the PDP context after timeout of t_T3396_UpperBound.

Please refer to the screenshot below.

	Source of change
	UTRAN34_SM_NIMTC.ttcn

	
	


Before:
	function fl_TC_11_1_3_4_Body() runs on UTRAN_PTC
  {
    var template (value) QualityOfService v_QualityOfService := f_UTRAN_InitialiseQoS(utran34_CellA, cell_DCH_64kPS_RAB_SRB);
    var InitialUE_Identity v_InitialUE_Id;
    var AccessStratumReleaseIndicator v_UE_ReleaseIndicator := tsc_UTRAN_Release99;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var ACTIVATEPDPCONTEXTREQUESTul v_ActivatePDPContextRequest;        
    var TI v_TI_R, v_TI_S;
    var boolean v_IPv4_Preferred := true;
    timer t_T3396_UpperBound := 10.0 * 1.1; // 10 sec + 10 % tolerance
    timer t_T3396_LowerBound := 10.0 * 0.9; // 10 sec - 10 % tolerance
    //@siclog "Step 1" siclog@
    // Initiate a context activation
    f_UT_ActivatePDPContextUG(UT, 1, v_QualityOfService);
    // Activate DHCP and ICMPv6
    f_UTRAN_IP_Handling_Start_Configuration(PDN_1, cell_DCH_64kPS_RAB_SRB);
    //@siclog "Step 1A" siclog@
    // The SS checks that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message
    // is set to "Delay Tolerant Access" //@sic R5-140633 sic@
    U_TM.receive(car_RRC_ConnReq(utran34_CellA, tsc_RB0, cr_108_RRC_ConnReq_r8(delayTolerantAccess))) -> value v_RLC_TR_DATA_IND;
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 1A");
    v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
    // Get UE release in RRC CONNECTION REQUEST and continue IdleUpdated using this release
    if (ispresent(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions)) {
      v_UE_ReleaseIndicator := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator;
    }
    f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
    // The UE transmits the SERVICE REQUEST message, service type = "signalling"
    f_UTRAN_GMM_ServReq(ps_domain, tsc_SeviceTypeSignalling);
    // The SS starts integrity protection
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellA, true, ps_domain);
    //@siclog "Step 2" siclog@
    //Request a PDP context activation.
    v_ActivatePDPContextRequest := f_UTRAN_PS_SessionAct_ReceivePDP_Req();
    // Check Device properties
    f_UTRAN34_DeviceProperties_CheckLowPriority(v_ActivatePDPContextRequest.deviceProperties, '1'B);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    //@siclog "Step 3" siclog@
    //The SS includes the IEs SM Cause with value #26 "insufficient resources" and T3396 with value 10 seconds.
    v_TI_R := v_ActivatePDPContextRequest.ti;    
    f_UTRAN34_CheckTI(v_TI_R);
    //Set TI when a message is sent from the network as a response to a request from UE
    v_TI_S.tiFlag :='1'B;
    v_TI_S.tiVal :=v_TI_R.tiVal;
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ActivatePDPContextReject(cs_ActPDPContextReject(v_TI_S,
                                                                                  '1A'O, //  value #26 "insufficient resources"
                                                                                  cs_GPRS_Timer2_3_IEI('37'O, tsc_Gprs3TimerUnit_2sec, '00101'B))))); // Set to 10 sec @sic R5s150329 Baseline Moving sic@
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //The SS verifies that the UE does not initiate any communication before the T3396 timer has expired
    t_T3396_LowerBound.start;

    t_T3396_UpperBound.start;

    alt {

      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,

                                      tsc_RB0,

                                      cr_RRC_RrcConnReqAny))

        {

          //RRC Connection Request message should not be received within the lower bound

          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4. RRC Connection Request message should not be received within the lower bound");

        }

      [] t_T3396_LowerBound.timeout {

        alt {

          [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,

                                          tsc_RB0,

                                          cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND

            {

              t_T3396_UpperBound.stop;

              f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 4");

              v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;

              f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);

            }

          [] t_T3396_UpperBound.timeout

            {

              //upper bound timer time-out

              f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");

            }

        }

      }

    }

    // The UE transmits the SERVICE REQUEST message, service type = "signalling"
    f_UTRAN_GMM_ServReq(ps_domain, tsc_SeviceTypeSignalling);
    // The SS starts integrity protection
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellA, true, ps_domain);    
    //@siclog "Steps 4-5" siclog@
    // Activate PDP context and RAB
    if (pc_IPv6 and (not pc_IPv4)) {
      v_IPv4_Preferred := false;
    }
    f_UTRAN34_EstablishPDP_Context(utran34_CellA, v_QualityOfService, v_UE_ReleaseIndicator, v_IPv4_Preferred);
  } // fl_TC_11_1_3_4_Body
}


After:

	function fl_TC_11_1_3_4_Body() runs on UTRAN_PTC
  {
    var template (value) QualityOfService v_QualityOfService := f_UTRAN_InitialiseQoS(utran34_CellA, cell_DCH_64kPS_RAB_SRB);
    var InitialUE_Identity v_InitialUE_Id;
    var AccessStratumReleaseIndicator v_UE_ReleaseIndicator := tsc_UTRAN_Release99;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var ACTIVATEPDPCONTEXTREQUESTul v_ActivatePDPContextRequest;        
    var TI v_TI_R, v_TI_S;
    var boolean v_IPv4_Preferred := true;
    timer t_T3396_UpperBound := 10.0 * 1.1; // 10 sec + 10 % tolerance
    timer t_T3396_LowerBound := 10.0 * 0.9; // 10 sec - 10 % tolerance
    //@siclog "Step 1" siclog@
    // Initiate a context activation
    f_UT_ActivatePDPContextUG(UT, 1, v_QualityOfService);
    // Activate DHCP and ICMPv6
    f_UTRAN_IP_Handling_Start_Configuration(PDN_1, cell_DCH_64kPS_RAB_SRB);
    //@siclog "Step 1A" siclog@
    // The SS checks that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message
    // is set to "Delay Tolerant Access" //@sic R5-140633 sic@
    U_TM.receive(car_RRC_ConnReq(utran34_CellA, tsc_RB0, cr_108_RRC_ConnReq_r8(delayTolerantAccess))) -> value v_RLC_TR_DATA_IND;
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 1A");
    v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
    // Get UE release in RRC CONNECTION REQUEST and continue IdleUpdated using this release
    if (ispresent(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions)) {
      v_UE_ReleaseIndicator := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator;
      fl_SetTestCaseRelease(utran34_CellA, v_UE_ReleaseIndicator);//WA#11_1_3_4
    }
    f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
    // The UE transmits the SERVICE REQUEST message, service type = "signalling"
    f_UTRAN_GMM_ServReq(ps_domain, tsc_SeviceTypeSignalling);
    // The SS starts integrity protection
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellA, true, ps_domain);
    //@siclog "Step 2" siclog@
    //Request a PDP context activation.
    v_ActivatePDPContextRequest := f_UTRAN_PS_SessionAct_ReceivePDP_Req();
    // Check Device properties
    f_UTRAN34_DeviceProperties_CheckLowPriority(v_ActivatePDPContextRequest.deviceProperties, '1'B);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    //@siclog "Step 3" siclog@
    //The SS includes the IEs SM Cause with value #26 "insufficient resources" and T3396 with value 10 seconds.
    v_TI_R := v_ActivatePDPContextRequest.ti;    
    f_UTRAN34_CheckTI(v_TI_R);
    //Set TI when a message is sent from the network as a response to a request from UE
    v_TI_S.tiFlag :='1'B;
    v_TI_S.tiVal :=v_TI_R.tiVal;
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ActivatePDPContextReject(cs_ActPDPContextReject(v_TI_S,
                                                                                  '1A'O, //  value #26 "insufficient resources"
                                                                                  cs_GPRS_Timer2_3_IEI('37'O, tsc_Gprs3TimerUnit_2sec, '00101'B))))); // Set to 10 sec @sic R5s150329 Baseline Moving sic@
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //The SS verifies that the UE does not initiate any communication before the T3396 timer has expired
    /*
    t_T3396_LowerBound.start;
    t_T3396_UpperBound.start;
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4. RRC Connection Request message should not be received within the lower bound");
        }
      [] t_T3396_LowerBound.timeout {
        alt {
          [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                          tsc_RB0,
                                          cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
            {
              t_T3396_UpperBound.stop;
              f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 4");
              v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
              f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
            }
          [] t_T3396_UpperBound.timeout
            {
              //upper bound timer time-out
              f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
            }
        }
      }
    }
    */
    //WA#11_1_3_4 : Change start
    t_T3396_UpperBound.start;
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4. RRC Connection Request message should not be received within the lower bound");
        }
      [] t_T3396_UpperBound.timeout
         {
           //upper bound timer time-out
           f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 4 : No RRC Connection received within T3396 timeout");
         }
    }
    f_UT_ActivatePDPContextUG(UT, 1, v_QualityOfService);
    U_TM.receive(car_RRC_ConnReq(utran34_CellA, tsc_RB0, cr_108_RRC_ConnReq_r8(delayTolerantAccess))) -> value v_RLC_TR_DATA_IND;
    v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
    // Get UE release in RRC CONNECTION REQUEST and continue IdleUpdated using this release
    if (ispresent(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions)) {
      v_UE_ReleaseIndicator := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator;
    }
    f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
    // The UE transmits the SERVICE REQUEST message, service type = "signalling"
    f_UTRAN_GMM_ServReq(ps_domain, tsc_SeviceTypeSignalling);
    // The SS starts integrity protection
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellA, true, ps_domain);    
    //@siclog "Steps 4-5" siclog@
    // Activate PDP context and RAB
    if (pc_IPv6 and (not pc_IPv4)) {
      v_IPv4_Preferred := false;
    }
    f_UTRAN34_EstablishPDP_Context(utran34_CellA, v_QualityOfService, v_UE_ReleaseIndicator, v_IPv4_Preferred);
  } // fl_TC_11_1_3_4_Body
}


4.6 f_UTRAN34_RB_SetUp_PS_RAB_WithR7
	Record Name
	f_UTRAN34_RB_SetUp_PS_RAB_WithR7

	Reason for change
	This function is called fl_TC_11_1_3_4_Body()->f_UTRAN34_EstablishPDP_Context-> f_UTRAN34_RB_SetUp_PS_RAB_WithR7.

In the current TTCN implementation of this function, the testcase flow goes to fl_TC_11_1_3_4_Body()->f_UTRAN34_EstablishPDP_Context-> f_UTRAN34_RB_SetUp_PS_RAB_WithR7-> f_UTRAN_RB_SetUp_PS_RAB.

The function f_UTRAN_RB_SetUp_PS_RAB handles the RB Setup Complete message within its body. The problem occurs when the flow comes out of this function, and in the function f_UTRAN34_RB_SetUp_PS_RAB_WithR7 the TTCN again waits for another RB SetupComplete!!  This is incorrect, and needs to be handled correctly.

	Summary of change
	In the function f_UTRAN34_RB_SetUp_PS_RAB_WithR7 created a boolean variable called flag, and set its defult value to true. Modified the TTCN to wait for RB Setupcomplete based on the value of this variable.

If the TTCN execution goes into the case where function  f_UTRAN_RB_SetUp_PS_RAB is being called, then flag is set as false because the RB SetupComplete is handled within this function.
Please refer to the screenshot below.

	Source of change
	UTRAN34_PDCP_Functions.ttcn

	
	


Before:
	function f_UTRAN34_RB_SetUp_PS_RAB_WithR7(UTRAN_CellId_Type p_CellId,
                                      RB_ConfigType     p_RbType,
                                      boolean           p_WaitForSetupCmpl := true,
                                      boolean           p_LoopbackRLC_PDU :=false,
                                      bitstring         p_RAB_Id := tsc_RAB_DefPS)
    runs on UTRAN_PTC
  {
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var UTRAN_CFN_Info_Type v_CfnInfo := f_CalculateActTime(p_CellId);
    var template (value) SS_ActivationTime v_SS_ActivationTime := cs_ActivateCFN(v_CfnInfo.actTime);    
    select (p_RbType) {
      case (cell_DCH_64kPS_RAB_SRB) {
        select (f_UTRAN_ConfigurationInfo_TestCaseRelease_Get()) {
          case(FDDr7) {
            if (not p_WaitForSetupCmpl) {
                    // Estimate the start value since we cannot take it from the Setup Complete message
                    v_SecurityInfo.start_PS := f_CalculateStartValue_ConnectedMode(v_SecurityInfo.start_PS);
                    f_UTRAN_Security_Set(v_SecurityInfo);
            }
            // Send RB_Set Up for DCH 64k PS for FDD in Rel-7
            f_UTRAN_RB_SetUp_Send(cs_108_RB_SetUp64k_PS_FDDr7(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                              v_CfnInfo.actTime,
                                                              p_RAB_Id,
                                                              f_UTRAN_CellInfo_GetULScrmCode_FDD(p_CellId),
                                                              f_UTRAN_CellInfo_GetPriScrmCode_FDD(p_CellId)));
            //Configure 64k PS RAB at SS in Rel-7
            f_SS_Config_64kPS_RAB_r7(p_CellId, v_SS_ActivationTime, p_LoopbackRLC_PDU);
          }
          case (TDDr7) {
            if (not p_WaitForSetupCmpl) {
                    // Estimate the start value since we cannot take it from the Setup Complete message
                    v_SecurityInfo.start_PS := f_CalculateStartValue_ConnectedMode(v_SecurityInfo.start_PS);
                    f_UTRAN_Security_Set(v_SecurityInfo);
            }
            f_UTRAN_RB_SetUp_Send(cs_108_RB_SetUp64k_PS_TDDr7(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                              v_CfnInfo.actTime,
                                                              p_RAB_Id,
                                                              f_UTRAN_CellInfo_GetCellParametersID_TDD(p_CellId)));
            //Configure 64k PS RAB at SS in Rel-7
            f_SS_Config_64kPS_RAB_r7(p_CellId, v_SS_ActivationTime, p_LoopbackRLC_PDU);
          }
          case else {
            f_UTRAN_RB_SetUp_PS_RAB ( p_CellId, p_RbType, p_WaitForSetupCmpl, p_LoopbackRLC_PDU);
            }
        }
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid configuration");
      }
    }
    if (p_WaitForSetupCmpl) {
        f_UTRAN_RRC_ReceiveRB_SetupCmpl_PS_RAB(p_CellId, p_RbType);
        }
      else {
        // just apply the ciphering at SS using the estimated start value
        f_UTRAN_SS_CipheringAM_RAB_UL_DL(p_CellId, ps_domain, p_RbType, v_SecurityInfo.authKeys.AuthPS.CK);
        }
   }
  }


After:

	function f_UTRAN34_RB_SetUp_PS_RAB_WithR7(UTRAN_CellId_Type p_CellId,
                                      RB_ConfigType     p_RbType,
                                      boolean           p_WaitForSetupCmpl := true,
                                      boolean           p_LoopbackRLC_PDU :=false,
                                      bitstring         p_RAB_Id := tsc_RAB_DefPS)
    runs on UTRAN_PTC
  {
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var UTRAN_CFN_Info_Type v_CfnInfo := f_CalculateActTime(p_CellId);
    var template (value) SS_ActivationTime v_SS_ActivationTime := cs_ActivateCFN(v_CfnInfo.actTime);
    var boolean flag := true;//WA#11_1_3_4
    select (p_RbType) {
      case (cell_DCH_64kPS_RAB_SRB) {
        select (f_UTRAN_ConfigurationInfo_TestCaseRelease_Get()) {
          case(FDDr7) {
            if (not p_WaitForSetupCmpl) {
                    // Estimate the start value since we cannot take it from the Setup Complete message
                    v_SecurityInfo.start_PS := f_CalculateStartValue_ConnectedMode(v_SecurityInfo.start_PS);
                    f_UTRAN_Security_Set(v_SecurityInfo);
            }
            // Send RB_Set Up for DCH 64k PS for FDD in Rel-7
            f_UTRAN_RB_SetUp_Send(cs_108_RB_SetUp64k_PS_FDDr7(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                              v_CfnInfo.actTime,
                                                              p_RAB_Id,
                                                              f_UTRAN_CellInfo_GetULScrmCode_FDD(p_CellId),
                                                              f_UTRAN_CellInfo_GetPriScrmCode_FDD(p_CellId)));
            //Configure 64k PS RAB at SS in Rel-7
            f_SS_Config_64kPS_RAB_r7(p_CellId, v_SS_ActivationTime, p_LoopbackRLC_PDU);
          }
          case (TDDr7) {
            if (not p_WaitForSetupCmpl) {
                    // Estimate the start value since we cannot take it from the Setup Complete message
                    v_SecurityInfo.start_PS := f_CalculateStartValue_ConnectedMode(v_SecurityInfo.start_PS);
                    f_UTRAN_Security_Set(v_SecurityInfo);
            }
            f_UTRAN_RB_SetUp_Send(cs_108_RB_SetUp64k_PS_TDDr7(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                              v_CfnInfo.actTime,
                                                              p_RAB_Id,
                                                              f_UTRAN_CellInfo_GetCellParametersID_TDD(p_CellId)));
            //Configure 64k PS RAB at SS in Rel-7
            f_SS_Config_64kPS_RAB_r7(p_CellId, v_SS_ActivationTime, p_LoopbackRLC_PDU);
          }
          case else {
            f_UTRAN_RB_SetUp_PS_RAB ( p_CellId, p_RbType, p_WaitForSetupCmpl, p_LoopbackRLC_PDU);
            flag := false;//WA#11_1_3_4
          }
        }
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid configuration");
      }
    }
    if (flag)//WA#11_1_3_4
    {
      if (p_WaitForSetupCmpl) {
        f_UTRAN_RRC_ReceiveRB_SetupCmpl_PS_RAB(p_CellId, p_RbType);
        }
      else {
        // just apply the ciphering at SS using the estimated start value
        f_UTRAN_SS_CipheringAM_RAB_UL_DL(p_CellId, ps_domain, p_RbType, v_SecurityInfo.authKeys.AuthPS.CK);
        }
   }
  }


5 Branches executed in test case 11.1.3.4
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Intel 7260 device
The Intel 7260 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_11_1_3_4_Intel_7260.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
7 References

